

## COMPILADORES E INTÉRPRETES

Proyecto  
Reglas de Sintaxis de MINIJAVA  
Segundo Cuatrimestre de 2023

### 1 Introducción

Este documento describe las reglas de sintaxis de MINIJAVA. Como se menciona en el documento de especificación del proyecto, MINIJAVA es un lenguaje que puede verse como una simplificación de Java, donde por una parte no se consideran elementos avanzados como Genericidad, Excepciones o Hilos, y por otra parte la estructura de su sintaxis es más estricta. En este documento se presentará la estructura sintáctica del lenguaje mediante su gramática. A partir de las producciones de esta gramática se podrán identificar con mayor facilidad aquellos elementos que alejan a MINIJAVA de Java.

### 2 Gramática de MiniJava

Cualquier programa MINIJAVA sintácticamente válido debe ser producto de la gramática que se presenta en esta sección. La gramática sigue la notación BNF-extendida, donde:

<b>terminal</b>	es un símbolo terminal
<b>&lt;Clase&gt;</b>	es un símbolo no terminal (además la primer letra es una mayúscula)
$\epsilon$	representa la cadena vacía
<b>&lt;X&gt;::=<math>\alpha</math></b>	representa una producción, con $\alpha$ una secuencia de terminales y no terminales
<b>&lt;X&gt;::=<math>\alpha \beta</math></b>	es una abreviación de <b>&lt;X&gt;::=<math>\alpha</math></b> y <b>&lt;X&gt;::=<math>\beta</math></b>

#### Producciones BNF

El no terminal de inicio es **<Inicial>** y las producciones de la Gramática de MINIJAVA son:

```
<Inicial> ::= <ListaClases>

<ListaClases> ::= <Clase> <ListaClases> |  $\epsilon$ 

<Clase> ::= <ClaseConcreta> | <Interface>

<ClaseConcreta> ::= class idClase <HerenciaOpcional> { <ListaMiembros> }

<Interface> ::= interface idClase <ExtiendeOpcional> { <ListaEncabezados> }

<HerenciaOpcional> ::= <HeredaDe> | <ImplementaA> |  $\epsilon$ 

<HeredaDe> ::= extends idClase

<ImplementaA> ::= implements idClase

<ExtiendeOpcional> ::= extends idClase |  $\epsilon$ 
```

```

<ListaMiembros> ::= <Miembro> <ListaMiembros> | ε

<ListaEncabezados> ::= <EncabezadoMetodo> <ListaEncabezados> | ε

<Miembro> ::= <Atributo> | <Metodo> | <Constructor>

<Atributo> ::= <EstaticoOpcional> <TipoMiembro> idMetVar ;

<Metodo> ::= <EstaticoOpcional> <TipoMiembro> idMetVar <ArgsFormales> <Bloque>

<EncabezadoMetodo> ::= <EstaticoOpcional> <TipoMiembro> idMetVar <ArgsFormales> ;

<Constructor> ::= public idClase <ArgsFormales> <Bloque>

<TipoMiembro> ::= <Tipo> | void

<Tipo> ::= <TipoPrimitivo> | idClase

<TipoPrimitivo> ::= boolean | char | int

<EstaticoOpcional> ::= static | ε

<ArgsFormales> ::= ( <ListaArgsFormalesOpcional> )

<ListaArgsFormalesOpcional> ::= <ListaArgsFormales> | ε

<ListaArgsFormales> ::= <ArgFormal>
<ListaArgsFormales> ::= <ArgFormal> , <ListaArgsFormales>

<ArgFormal> ::= <Tipo> idMetVar

<Bloque> ::= { <ListaSentencias> }

<ListaSentencias> ::= <Sentencia> <ListaSentencias> | ε

<Sentencia> ::= ;
<Sentencia> ::= <Asignacion> ;
<Sentencia> ::= <Llamada> ;
<Sentencia> ::= <VarLocal> ;
<Sentencia> ::= <Return> ;
<Sentencia> ::= <If>
<Sentencia> ::= <While>
<Sentencia> ::= <Bloque>

<Asignacion> ::= <Expresion>

<llamada> ::= <Expresion>

<VarLocal> ::= var idMetVar = <ExpresionCompuesta>

<Return> ::= return <ExpresionOpcional>

<ExpresionOpcional> ::= <Expresion> | ε

<If> ::= if ( <Expresion> ) <Sentencia>
<If> ::= if ( <Expresion> ) <Sentencia> else <Sentencia>

```

```

<While> ::= while ( <Expresion> ) <Sentencia>

<Expresion> ::= <ExpresionCompuesta> = <Expresion>
<Expresion> ::= <ExpresionCompuesta>

<ExpresionCompuesta> ::= <ExpresionCompuesta> <OperadorBinario> <ExpresionBasica>
<ExpresionCompuesta> ::= <ExpresionBasica>

<OperadorBinario> ::= || | && | == | != | < | > | <= | >= | + | - | * | / | %

<ExpresionBasica> ::= <OperadorUnario> <Operando>
<ExpresionBasica> ::= <Operando>

<OperadorUnario> ::= + | - | !

<Operando> ::= <Literal>
<Operando> ::= <Acceso>

<Literal> ::= null | true | false | intLiteral | charLiteral | stringLiteral

<Acceso> ::= <Primario> <EncadenadoOpcional>

<Primario> ::= <AccesoThis>
<Primario> ::= <AccesoVar>
<Primario> ::= <AccesoConstructor>
<Primario> ::= <AccesoMetodo>
<Primario> ::= <AccesoMetodoEstatico>
<Primario> ::= <ExpresionParentizada>

<AccesoThis> ::= this

<AccesoVar> ::= idMetVar

<AccesoConstructor> ::= new idClase <ArgsActuales>

<ExpresionParentizada> ::= ( <Expresion> )

<AccesoMetodo> ::= idMetVar <ArgsActuales>

<AccesoMetodoEstatico> ::= idClase . idMetVar <ArgsActuales>

<ArgsActuales> ::= ( <ListaExpsOpcional> )

<ListaExpsOpcional> ::= <ListaExps> | ε

<ListaExps> ::= <Expresion>
<ListaExps> ::= <Expresion> , <ListaExps>

<EncadenadoOpcional> ::= <VarEncadenada> | <MetodoEncadenado> | ε

<VarEncadenada> ::= . idMetVar <EncadenadoOpcional>

<MetodoEncadenado> ::= . idMetVar <ArgsActuales> <EncadenadoOpcional>

```